实证论文：债务期限结构影响因素研究

数据来源：RESSET 金融数据库(www.resset.com)。 

样本原则:
1.研究期限是2006-2012;
2.非金融类上市公司;
3.合并未调整报表数据;
4.只选择A股市场主板上市公司数据，踢除B股或H股公司，同时拥有A股和B股或A股和H股的公司也不在考虑范围之类;
5.样本踢除ST.和PT类上市公司。
基于上述选择标准，最后筛选出上海证劵交易所上市公司7年的数据作为样本。

注：财务数据中，分子缺失都记为0。
注：需要无缺失值的标识变量CsrcIcCd1  Csrcicnm1  Listedstate最后处理。
注：标红的为用2种方法计算的同一数据。

表1 计算公式与数据源设计
	
	序号
	变量名称
	对应程序变量名
	计算公式（多个数据来源时，在()内写明数据源）
	数据源

	样本公司
	
	公司代码；
证监会行业代码
	ComCd
CsrcIcCd1
	1）取非金融行业且只有A股上市的公司(CInfo): where Lstflg='A'  and FinIndFlg=0 ;  
2）取上交所数据，SecuMkt(LstkInfo)=83；
3）筛选2006-2012年 且 踢除ST.和PT类上市公司(YRESSTK)：listedstate='Norm' and '1jan2006'd<=Date<='31dec2012'd and  R_SecuCode ^=''；
最后得到样本公司SampCom
	公司信息CInfo；
最新股票信息LstkInfo；
年股票综合数据YRESSTK

	应变量
	1
	债务期限结构
	LiabilitiesTermStruct 
	债务期限结构=长期债务/总债务=长期负债合计(元)TotLTlia(Bs_Old) /负债合计 (元)Totlia(Bs_Old)
	非金融行业资产负债表_旧准则BS_Old

	自变量
	2
	信息不对称程度
	InfoAsymDegree
	信息不对称程度=ln(1+企业年限)= Log(1+Enddt-Estdt)
	公司信息CINFO

	
	3
	企业规模
	EntSize
	企业规模=总资产的自然对数=Log(Totass) (BS)
	非金融行业资产负债表BS；非金融行业利润表IS；非金融行业现金流量表SCF；
资产负债表附注: 固定资产及折旧 BSNotE_FixassDepr； 财务指标FinRatio；年市值YrMv。

	
	4
	财务杠杆
	FinLeverage
	财务杠杆=总负债/总资产 =Totlia/Totass (BS)
	

	
	5
	资产流动性
	AssLiquidity
	资产流动性=流动资产/流动负债 =Totcurass/Totcurlia (BS)
	

	
	6
	资产期限结构_BS计算
	AssTermStruct_BS
	资产期限结构_BS计算=固定资产/总资产=FixAss/TotAss (BS)
	

	
	7
	资产期限结构_BSIS折旧计算
	AssTermStruct_BSIS
	资产期限结构_BSIS折旧计算
=[设备资产期限FixAssTerm×（设备资产Fixass(BS)/总资产Totass(BS)]+[流动资产期限CurAssTerm×(流动资产Totcurass(BS)/总资产Totass(BS))]
FixAssTerm设备资产期限
=固定资产Fixass(BS)/折旧(BSNotE_FixassDepr中的累计折旧期末数Endaccudepr)
CurAssTerm流动资产期限
=流动资产Totcurass(BS) /营业费用Opeexp(IS)
	

	
	8
	总资产
	TotAss (BS)
	
	

	
	9
	成长性
	Growth
	成长性=(本年主营业务收入-本年年初主营业务收入)/本年年初主营业务收入                 
=(Incmope-Lag(Incmope))/Lag(Incmope) (IS)
	

	
	10
	自由现金流
	FreeCF
	自由现金流=税后利润+固定资产折旧-投资项目现金流出
=Netprf(IS)+(Endaccudepr-Opaccudepr)(BSNotE_FixassDepr)-NCFinv(SCF)
	

	
	11
	实际税率
	ActTaxRat
	实际税率=当前所得税额/税前利润=Incmtax/Prfbftax (IS)
	

	
	12
	非债务税盾
	NonDebtTaxShield
	非债务税盾=(固定资产折旧+无形资产摊销)*总资产 
=((Endaccudepr-Opaccudepr)+无形资产摊销)*总资产
=((Endaccudepr-Opaccudepr)+Amotintanass(SCF))*TotAss(BS)
	

	
	13
	市场价财务杠杆
	MktFinLeverage
	市场价财务杠杆=总负债Totlia(BS)/(股票的市场价值年末总市值–人民币计价_YrmcCNY_End (YrMv) +总负债Totlia(BS))
	

	
	14
	自由现金流销售比
	FreeCFIncmRatio
	自由现金流销售比=自由现金流自由现金流量 (元)_FreeCF（FinRatio）/销售收入（IS中的营业收入Incmope）
	

	
	15
	企业质量
	EntQuality
	企业质量=本年每股收益增加额/年末每股市价 =(EPS-lag(EPS))/Clpr
	年股票综合数据YRESSTK

	
	16
	行业管制(虚拟变量)
	IndCntrlID
	当公司属于管制行业时,IndCntrlID为1 ，否则为0;其中将D类（电力、煤气及水的生产和供应业）和F类（交通运输、仓储业）的企业视为”管制型企业”
	

	
	17
	上市状态
	ListedState
	
	

	
	18
	企业价值波动性_股票收益计算
	Dretvolatility_yr 
	企业价值波动性_股票收益计算=股票收益率的标准差Dretvolatility_yr
	日收益波动率_年度数据ResVar.Dretvolatility_yr

	
	19
	企业价值波动性_EBIT计算
	EntValVolatility_EBIT
	企业价值波动性_EBIT计算=EBIT变化百分比-EBIT变动百分比平均值=((EBIT-上年度EBIT)/上年度EBIT)-average( ((EBIT-上年度EBIT)/上年度EBIT)
=((EBIT-lag(EBIT))/lag(EBIT))-average( ((EBIT-lag(EBIT))/Lag(EBIT))
息税前利润EBIT=利润总额TotalPrf +利息费用
利息费用=应付利息IntPay -应收利息IntRecv
	财务指标与比率Finindratio_old

	
	20
	资产收益率
	ROA
	
	

	
	21
	托宾Q值
	QVal
	
	

	
	22
	第一大股东持股比例
	LargSH_HSPct
	第一大股东持股比例=股权集中度1 OwnCon1
	股权集中度OwnCon

	
	23
	少数大股东持股比例
	SmallSH_HSPct
	少数大股东持股比例=第二至第十大股东股份/总股本
=股权集中度10 OwnCon10-股权集中度1 OwnCon1
	

	
	24
	国家持股比例
	State_HSPct
	国家持股比例=国有股/总股本 =国有股比例StateShrPct/100 (数据源是百分比)
	

	
	25
	管理者持股比例
	ExeHoldRt
	管理者持股比例= 管理层持股总和/总股本
=公司每年期末持股数之和/ComFullShr
	管理层持股与薪酬ExeHold；公司股数变动历史CShrHis。

	
	26
	现金股利(虚拟变量)
	DividendID
	支付现金股利的公司=1. 不支付现金股利的公司=0。
	分配Dist

	
	27
	政府债利差
	IRDif
	政府债利差=长期债劵利率CoupRt (10年期)-短期债劵利率CoupRt (1年期)
	债券信息Bdinfo




数据处理与计算：根据表1的计算公式与数据源设计，完成所有变量的数据处理与计算。

样本公司：
/* 非金融公司，只有A股上市 */
data sampCom1;
set resstk.Cinfo;
where Lstflg='A'  and ( csrciccd1^='I' or comcd="C600643" ) ;  
keep comcd csrciccd1;
run;

/* 或直接用FinIndFlg */
data sampCom1;
set resstk.cinfo;
where Lstflg='A'  and FinIndFlg=0 ;  
keep comcd csrciccd1 ;
run;

proc sql;
create table SampCom2 as 
select distinct b.Comcd, b.csrciccd1
from SampCom1 a left join Resstk.lstkinfo b
on a.comcd=b.comcd
where SecuMkt=83;
quit;


/* 2006-2012年 且 踢除ST.和PT类上市公司 */
proc sql;
create table  temp.sampCom as 
select distinct b.comcd 
from sampCom2 a left join resstk.Yresstk b 
on a.comcd=b.comcd
where listedstate='Norm'  and  '1jan2006'd<=Date<='31dec2012'd and  R_SecuCode ^='' ;
quit;


1. 债务期限结构=长期债务/总债务
LiabilitiesTermStruct债务期限结构=长期负债合计(元)TotLTlia /负债合计 (元)Totlia 
注：债务期限结构是因变量所以不能为空。如果是空的就没有用。把空的筛选掉。
长期债务 =长期负债合计 (元) 
总债务  =负债合计 (元) 
数据源：非金融行业资产负债表_旧准则BS_Old 
原来计算的公式：
FinTermStruct 财务期限结构=长期债务/总债务=TotNcurlia/Totlia (BS)
(应该叫债务期限结构)。数据源使用的是BS。有缺失值。

Data LiaTermStruct1;
Set ResFin.BS_Old;
Where  (2005<=year(Enddt)<=2013 and ReportType='Q4' and ConFlg=1 and AdjFlg=0) ;  Year=year(Enddt);
Keep  Year  ComCd  TotLTlia  Totlia;
Run;

Proc sort data=LiaTermStruct1;
By comcd year;
Run;

Data LiaTermStruct2;
Set  LiaTermStruct1;
If TotLTlia=. Then TotLTlia=0;
If (Totlia^=. And Totlia^=0) then  LiabilitiesTermStruct= TotLTlia/ Totlia ;
Run;

Data LiaTermStruct3;
Set  LiaTermStruct2;
By comcd;
If first.comcd then delete ; /*每家公司的第一条Growth数据是与上家公司计算的，要删除 */
Run;

/*取样本公司数据 */
Proc  sql;
Create table  temp.LiaTermStruct  as
Select  b.ComCd,b.year,b.LiabilitiesTermStruct
From temp.sampCom  A  left  join  LiaTermStruct3  b
On a.comcd=b.comcd;
Quit;

proc sort data= temp.LiaTermStruct;
by Comcd Year;
run;


2. 信息不对称程度=ln（1+企业年限）
InfoAsymDegree 信息不对称程度=Log(1+企业年限) = Log(1+Enddt-Estdt) 
注：样本企业2006-2012年每年一条数据，企业年限=Yrdif(2006---2012年12月31- Estdt)
数据源: 公司信息CINFO
用sql直接计算取可：

Proc  sql;
Create table  InfoAsymDegree1  as
Select  b.comcd, b. Estdt 
From temp.sampCom  A  left  join  resstk.cinfo  b
On a.comcd=b.comcd;
Quit;

Data  InfoAsymDegree2;
Set InfoAsymDegree1;
Do year=2006  to 2012;
Enddt=mdy(12,31,year);
yrdif =yrdif(Estdt,Enddt,'act/act');
Output;
End;
Run;

Data temp.InfoAsymDegree;
Set InfoAsymDegree2;
InfoAsymDegree =Log(1+yrdif);
Label InfoAsymDegree ='InfoAsymDegree信息不对称程度';
keep comcd Year InfoAsymDegree;
run;

proc sort data=temp.InfoAsymDegree;
by comcd year;
run;

3. 企业规模=总资产的自然对数
EntSize 企业规模=总资产的自然对数=Log(Totass) (BS)
4. 财务杠杆=总负债/总资产
FinLeverage 财务杠杆=总负债/总资产 =Totlia/Totass (BS)
5. 资产流动性=流动资产/流动负债
AssLiquidity 资产流动性=流动资产/流动负债 =Totcurass/Totcurlia (BS)
6. 资产期限结构_BS计算=固定资产/总资产 
AssTermStruct_BS 资产期限结构_BS计算=固定资产/总资产=FixAss/TotAss (BS)
7. 资产期限结构_BSIS折旧计算
=[设备资产期限×（设备资产/总资产）]+[流动资产期限×(流动资产/总资产)]
设备资产期限=固定资产/折旧
流动资产期限=流动资产/营业费用
AssTermStruct_BSIS资产期限结构_BSIS折旧计算
=[设备资产期限FixAssTerm×（设备资产Fixass(BS)/总资产Totass(BS)]
+[流动资产期限CurAssTerm×(流动资产Totcurass(BS)/总资产Totass(BS))]
FixAssTerm设备资产期限
=固定资产Fixass(BS)/折旧(BSNotE_FixassDepr中的累计折旧期末数Endaccudepr)
CurAssTerm流动资产期限
=流动资产Totcurass(BS) /营业费用Opeexp(IS)
注：设备资产就是固定资产，营业费用就是销售费用
8.  总资产TotAss (BS)
9.   成长性=(本年主营业务收入-本年年初主营业务收入)/本年年初主营业务收入
Growth 成长性=(本年主营业务收入-本年年初主营业务收入)/本年年初主营业务收入 
                 =(Incmope-Lag(Incmope))/Lag(Incmope) (IS)
10.	自由现金流=税后利润+固定资产折旧-投资项目现金流出
FreeCF自由现金流=税后利润+固定资产折旧-投资项目现金流出 
=Netprf(IS)+(Endaccudepr-Opaccudepr)(BSNotE_FixassDepr)-NCFinv(SCF)
11.	实际税率=当前所得税额/税前利润
ActTaxRat 实际税率=当前所得税额/税前利润=Incmtax/Prfbftax (IS)
12.	非债务税盾=（固定资产折旧+无形资产摊销）/总资产
NonDebtTaxShield 非债务税盾=(固定资产折旧+无形资产摊销)*总资产 
                 =((Endaccudepr-Opaccudepr)+无形资产摊销)*总资产
             =((Endaccudepr-Opaccudepr)+Amotintanass(SCF))*TotAss(BS)
13.	自由现金流销售比=自由现金流自由现金流量 (元) /销售收入
FreeCFIncmRatio 自由现金流销售比
=自由现金流自由现金流量 (元)_FreeCF（FinRatio）/销售收入（IS中的营业收入Incmope）
14.	市场价财务杠杆=总负债/(股票的市场价值+总负债)
MktFinLeverage市场价财务杠杆=总负债Totlia(BS)/(股票的市场价值年末总市值–人民币计价_YrmcCNY_End (YrMv) +总负债Totlia(BS))
注：股票的市场价值=上市股票的总市值 
3-14数据源：非金融行业资产负债表BS；非金融行业利润表IS；非金融行业现金流量表SCF；
资产负债表附注: 固定资产及折旧 BSNotE_FixassDepr；财务指标FinRatio；年市值YrMv。

Data bs;
Set ResFin.BS;
Where  (2005<=year(Enddt)<=2013 and ReportType='Q4' and ConFlg=1 and AdjFlg=0) ;  /*绝对不要加新会计准则 AccStd=1控制，不然就没有2007年前的数据了。 */
Year=year(Enddt);
Keep Year  comcd    TotNcurlia Totlia  Totass  Totcurass  Totcurlia  FixAss  ;
Run;

Proc sort data=BS;
By comcd year;
Run;

Data Is;
Set ResFin.IS;
Where  (2005<=year(Enddt)<=2013 and ReportType='Q4' and ConFlg=1 and AdjFlg=0) ;  /*绝对不要加新会计准则 AccStd=1控制，不然就没有2007年前的数据了。 */
Year=year(Enddt);
Keep Year  comcd   Incmope  Netprf  Incmtax  Prfbftax  Opeexp;
Run;

Proc sort data=IS;
By comcd year;
Run;

Data SCF;
Set ResFin.SCF;
Where  (2005<=year(Enddt)<=2013 and ReportType='Q4' and ConFlg=1 and AdjFlg=0) ;  /*绝对不要加新会计准则 AccStd=1控制，不然就没有2007年前的数据了。 */
Year=year(Enddt);
Keep Year  comcd   NCFinv  Amotintanass;
Run;

Proc sort data=SCF;
By comcd year;
Run
;
Data  BSNotE_FixassDepr;
Set ResFin.BSNotE_FixassDepr;
Where  (2005<=year(Enddt)<=2013 and ReportType='Q4' and ConFlg=1 and AdjFlg=0 and Fixasstype=99) ;  /*绝对不要加新会计准则 AccStd=1控制，不然就没有2007年前的数据了。Fixasstype=99是合计数 */
Year=year(Enddt);
Keep Year  comcd   Endaccudepr Opaccudepr;
Run;

Proc sort data= BSNotE_FixassDepr;
By comcd year;
Run;

Data FinRatio;
Set ResFin.FinRatio;
Where  (2005<=year(Enddt)<=2013 and ReportType='Q4' ) ;  
Year=year(Enddt);
Keep  Year  ComCd  FreeCF;
Run;

Proc sort data= FinRatio;
By comcd year;
Run;

Data temp.data1;
Merge  BS  IS  Scf  BSNotE_FixassDepr FinRatio;
By comcd year;
EntSize= log(Totass);
FinLeverage=sum( Totlia,0)/Totass;
AssLiquidity= sum(Totcurass,0)/Totcurlia;
AssTermStruct_BS=sum(FixAss,0)/TotAss;
FixAssTerm= sum(Fixass,0)/Endaccudepr;
CurAssTerm=sum(Totcurass,0)/Opeexp;
AssTermStruct_BSIS =(FixAssTerm*sum(Fixass,0)/Totass)+(CurAssTerm*sum(Totcurass,0)/ Totass);

Growth=(Incmope-Lag(Incmope))/Lag(Incmope);
FreeCF=sum(Netprf,Endaccudepr,-Opaccudepr)-sum(NCFinv,0);
ActTaxRat= sum(Incmtax,0)/Prfbftax ;
NonDebtTaxShield=sum(Endaccudepr,-Opaccudepr, Amotintanass)*sum(TotAss,0);
FreeCFIncmRatio = sum(FreeCF,0)/ Incmope;
Keep  Year  comcd     
 EntSize  FinLeverage  AssLiquidity  AssTermStruct_BS   AssTermStruct_BSIS 
Growth FreeCF ActTaxRat NonDebtTaxShield Totass FreeCFIncmRatio Totlia
;
Label 
EntSize='EntSize企业规模'
FinLeverage='FinLeverage财务杠杆'
AssLiquidity='AssLiquidity资产流动性'
AssTermStruct_BS='AssTermStruct_BS资产期限结构_BS计算' 
AssTermStruct_BSIS='AssTermStruct_BSIS资产期限结构_BSIS折旧计算'
Growth='Growth成长性'
FreeCF='FreeCF自由现金流'
ActTaxRat='ActTaxRat实际税率'
NonDebtTaxShield='NonDebtTaxShield非债务税盾' 
FreeCFIncmRatio='FreeCFIncmRatio自由现金流销售比'
;
Run;

Data temp.data1;
Set  temp.data1;
By comcd;
If first.comcd then Growth=. ; /*每家公司的第一条Growth数据是与上家公司计算的，要删除 */
Run;

/*取样本公司数据 */
Proc  sql;
Create table  temp.data1  as
Select  b.* 
From temp.sampCom  A  left  join  temp.data1  b
On a.comcd=b.comcd;
Quit;

proc sort data= temp.data1;
by Comcd Year;
run;

Data YrMv;
Set ResMv.Yrmv;
Where  (2006<=year(Date)<=2012) ;  
Year=year(Date);
Keep  Year  ComCd  YrtmvCNY_end;
Run;

Proc  sql;
Create table  YrMv1  as
Select  * 
From temp.sampCom  A  left  join  YrMv  b
On a.comcd=b.comcd
Where year is not missing;
Quit;

Proc sort data= YrMv1;
By comcd year;
Run;

Data temp.data1;
Merge temp.data1 YrMv1;
MktFinLeverage=Totlia/(YrtmvCNY_end +Totlia) ;
By comcd year;
Label MktFinLeverage='MktFinLeverage市场价财务杠杆';
Drop YrtmvCNY_end Totlia;
Run;


15.	企业质量=本年每股收益增加额/年末每股市价
EntQuality企业质量=本年每股收益增加额/年末每股市价 =(EPS-lag(EPS))/Clpr
16.	行业管制（虚拟变量）：当公司属于管制行业时为1 ，否则为0。
IndCntrlID 行业管制（虚拟变量）,当公司属于管制行业时,IndCntrlID为1 ，否则为0;其中将D类（电力、煤气及水的生产和供应业）和F类（交通运输、仓储业）的企业视为”管制型企业”
Keep 相关行业分类，用于最后计算行业管制IndCntrlID。
Keep comcd Year CsrcICCd CsrcIcNm CsrcIcCd1 CsrcIcNm1 Stkcd Lstknm Stkcdotrd listedstate;
17.	上市状态listedstate
15-17数据源:　年股票综合数据YRESSTK
用sql直接计算取可：

data EntQuality1;
set resstk.YRESSTK;
lag_EPS= lag(EPS) ;
EntQuality=(EPS- lag_EPS)/Clpr;
Year=year(date);
Keep comcd Year CsrcICCd CsrcIcNm CsrcIcCd1 CsrcIcNm1 Stkcd Lstknm Stkcdotrd listedstate  EntQuality ;  /* Stkcd Lstknm Stkcdotrd listedstate 在这里保留最准确 */
If listedstate='Norm' and 2005<=year<=2012  and  R_SecuCode ^='' ;
By stkcd ;
If first.stkcd then EntQuality=.; /*每家公司的第一条数据是与上家公司计算的，要删除 */
Run;

Proc  sql;
Create table  temp.EntQuality  as
Select  b.* 
From temp.sampCom  A  left  join  EntQuality1  b
On a.comcd=b.comcd;
Quit;

Proc sort data= temp.EntQuality;
By comcd year;
Run;


18.	企业价值波动性_股票收益计算=股票收益率的标准差
Dretvolatility_yr企业价值波动性_股票收益计算=股票收益率的标准差Dretvolatility_yr。
数据源：日收益波动率_年度数据ResVar.Dretvolatility_yr。已计算好。

Proc  sql;
Create table  temp.Dretvolatility_yr  as
Select  b.ComCd,b.year,b.Dretvolatility_yr 
From temp.sampCom  A  left  join  resvar.Dretvolatility_yr   b
On a.comcd=b.comcd;
Quit;

Proc sort data= temp.Dretvolatility_yr;
By comcd year;
Run;


19.	企业价值波动性_EBIT计算=EBIT变化百分比-EBIT变动百分比平均值
EntValVolatility_EBIT 企业价值波动性_EBIT计算=EBIT变化百分比-EBIT变动百分比平均值 
    =((EBIT-上年度EBIT)/上年度EBIT)-average( ((EBIT-上年度EBIT)/上年度EBIT)
    =((EBIT-lag(EBIT))/lag(EBIT))-average( ((EBIT-lag(EBIT))/Lag(EBIT))
息税前利润EBIT=利润总额TotalPrf +利息费用
利息费用=应付利息IntPay -应收利息IntRecv
注：average指的是样本均值  
20.	资产收益率（ROA）
21.	托宾Q值（QVal）
19-21数据源: 财务指标与比率Finindratio_old　(单独使用Sql方便计算按年分组的均值)

/* 
Data FinInd;
Set ResFin.FinInd;
EBIT=TotalPrf+IntPay-IntRecv;
EBITchgRat=sum(EBIT,-lag(EBIT))/lag(EBIT);
where 2005<=year(Enddt)<=2013 and ReportType='Q4' and Adjflg=0;
Year=year(Enddt);
Keep year comcd  EBITchgRat;
run;
*/
/* Finind中的数据缺失太多 */

Data EntValVolatility1;
Set ResFin.Finindratio_old ;
Lag_EBIT= lag(EBIT);
EBITchgRat=sum(EBIT,- Lag_EBIT)/ Lag_EBIT;
where 2005<=year(Enddt)<=2013 and ReportType='Q4';
Year=year(Enddt);
Keep year comcd EBIT EBITchgRat ROA QVal ;
By comcd;
If first.comcd then EBITchgRa=.; /*每家公司的第一条数据是与上家公司计算的，要删除 */
run;


Proc  sql;
Create table temp.EntValVolatility as
Select  Comcd, Year, EBITchgRat , avg(EBITchgRat)  as  avgEBITchgRat,
(EBITchgRat-calculated avgEBITchgRat) as EntValVolatility_EBIT ,ROA,QVal 
From EntValVolatility1
group by year; 
quit;

/* 选择样本公司数据 */
Proc  sql;
Create table  temp.EntValVolatility as
Select  b.* 
From temp.sampCom  A  left  join  temp.EntValVolatility  b
On a.comcd=b.comcd;
Quit;

proc sort data= temp.EntValVolatility;
by Comcd Year;
run;


22.	第一大股东持股=第一股东股份/总股本
LargSH_HSPct 第一大股东持股比例=股权集中度1 OwnCon1
注：持股比例=持股数/公司总股数。
23.	少数大股东持股=第二至第十大股东股份/总股本
SmallSH_HSPct 少数大股东持股比例=第二至第十大股东股份/总股本
=股权集中度10 OwnCon10-股权集中度1 OwnCon1
24.	国家持股=国有股/总股本
State_HSPct国家持股比例=国有股/总股本 =国有股比例StateShrPct/100 (数据源是百分比)
注：国有股比例StateShrPct =公司国有股/公司总股数*100。
22-24数据源：股权集中度OwnCon

Data temp.HSPct;
Set resstk. OwnCon;
Where Yrflg='1';
Year=year(Enddt);
Keep comcd year  LargSH_HSPct SmallSH_HSPct State_HSPct;
LargSH_HSPct= OwnCon1;
SmallSH_HSPct=(OwnCon10-OwnCon1);
State_HSPct= StateShrPct/100;
Label 
LargSH_HSPct='LargSH_HSPct第一大股东持股比例'
SmallSH_HSPct='SmallSH_HSPct少数大股东持股比例'
State_HSPct='State_HSPct国家持股比例'
;
If 2006<=year<=2012;
Run;

Proc  sql;
Create table  temp.HSPct  as
Select  b.* 
From temp.sampCom  A  left  join  temp.HSPct b
On a.comcd=b.comcd;
Quit;


Proc sort data= temp.HSPct;
By comcd year;
Run;

25.	管理者持股比例= 管理层持股总和/总股本 
ExeHoldRt管理者持股比例= 管理层持股总和/总股本
=公司每年期末持股数之和/ComFullShr
数据源：管理层持股与薪酬ExeHold；公司股数变动历史CShrHis。

data ExeHold1;
set resstk.ExeHold;
Year=year(EndDt);
Keep comcd Year ShrEnd  Yrflg;  
If 2006<=year<=2012 and Yrflg=1; /*取2006至2012年末（Yrflg=1）的期末持股数*/
Run;

Proc Sort data=ExeHold1;
By comcd year;
/*计算每个公司每年管理层期末持股数之和*/
Proc means sum  data=ExeHold1 noprint;;
Output out=ExeHold2  sum=sum;
by comcd year;
Var ShrEnd;
Run;

Data  CShrHis;
Set  resstk.CShrHis;
Year=year(DclrDt);
Keep comcd year   ComFullShr;
If 2006<=year<=2012;
Run;

Proc sort data=CShrHis;
By comcd year;
Run;

Data  CShrHis1;
Set  CShrHis;
by comcd year;
if last.year;
run;

Data ExeHold;
Merge  CShrHis1 ExeHold2;
By comcd year;
ExeHoldRt=sum(sum,0)/ComFullShr;
Keep comcd year ExeHoldRt;
where comcd is not missing and year is not missing;
Run;

Proc  sql;
Create table  temp. ExeHold  as
Select  b.* 
From temp.sampCom  A  left  join  ExeHold  b
On a.comcd=b.comcd;
Quit;

Proc sort data= temp. ExeHold;
By comcd year;
Run;

26.	现金股利虚拟变量 支付现金股利的公司=1. 不支付现金股利的公司=0 
DividendID现金股利(虚拟变量),支付现金股利的公司=1. 不支付现金股利的公司=0。
注：证鉴会把提出现金分红作为上市公司再筹资的必要条件，派现的公司为1，否则为0。
 数据源: 分配Dist

data Dividend1;
set resstk.Dist;
Year=year(date);
Keep comcd Year DividendID;  
If DividendFlg=1  then DividendID=1;
Else DividendID =0;
If 2006<=year<=2012;
Run;
Proc sort data= Dividend1;
By comcd Year DividendID;
Run;
/*由于一年中有多次分红，只要出现一次现金股利分红，那么DividendID =1*/
Data Dividend1;
Set Dividend1;
By comcd Year ;
If last.year;
Run;


Proc  sql;
Create table  temp.Dividend  as
Select  b.* 
From temp.sampCom  A  left  join  Dividend1  b
On a.comcd=b.comcd ;
Quit;

Proc sort data= temp.Dividend;
By comcd year;
Where comcd is not missing;
Run;

27.	政府债利差=长期债劵利率(10年期)-短期债劵利率(1年期)
IRDif政府债利差=长期债劵利率CoupRt (10年期)-短期债劵利率CoupRt (1年期)
数据源: 债券信息Bdinfo
 
data BdCoupRt1;
set resbd.bdinfo;
year=year(LstDt);
if BdType=4 and 2006<=year<=2012 and IntMd=1 and Maturity in (1,10) ;
keep BdType year CoupRt Maturity;
run;
/*删除重复项*/
proc sort data=BdCoupRt1 dupout=nodups3 nodupkey;
by year Maturity;
run;
/*计算利差*/
data temp.BdCoupRt;
set BdCoupRt1;
IRDif = CoupRt-lag(CoupRt);
By year;
If first.year then delete;
label 
IRDif='IRDif政府债利差';
Keep year IRDif ;
Run;



合并为一个大数据集：Results
Data results1;
Merge  temp.LiaTermStruct  temp.InfoAsymDegree  temp.data1 temp.Dretvolatility_yr temp.EntValVolatility temp.EntQuality  temp.HSPct temp.Dividend  temp. ExeHold  
;
By comcd year;
Label 
Dretvolatility_yr='Dretvolatility_yr企业价值波动性_股票收益计算'
EntValVolatility_EBIT='EntValVolatility_EBIT企业价值波动性_EBIT计算'
EntQuality='EntQuality企业质量'
DividendID='DividendID现金股利(虚拟变量)'
ExeHoldRt='ExeHoldRt管理层持股比例'
year='Year年'
LiabilitiesTermStruct='LiabilitiesTermStruct债务期限结构'
Listedstate='Listedstate上市状态'
;
Where 2006<=year<=2012;
If DividendID=. Then DividendID=0;
Run;



proc sql;
create table temp.results as
select a.*, b. IRDif  from results1 a right join temp.BdCoupRt b 
on a.year=b.year;
quit;


Proc sort data= temp.results;
By comcd year;
Run;

/*删除因变量LiabilitiesTermStruct='债务期限结构LiabilitiesTermStruct'缺失的观测 */
Data temp.results;
Set temp.results;
If LiabilitiesTermStruct=. Then delete;
Run;

/* 填充CsrcIcCd1 csrcicnm1  Listedstate */
data temp.results;
set  temp.results;
by comcd;
retain ;
if first. comcd then  do;
CsrcIcCd1_1= CsrcIcCd1;
csrcicnm1_1= csrcicnm1;
end;

else if CsrcIcCd1^=''  then  do;
CsrcIcCd1_1= CsrcIcCd1;
csrcicnm1_1= csrcicnm1;
End;

CsrcIcCd1= CsrcIcCd1_1;
csrcicnm1= csrcicnm1_1;
Drop CsrcIcCd1_1 csrcicnm1_1;
Listedstate='Norm';
Run;

Proc sort data= temp.results;
By comcd descending year;
Run;

data temp.results;
set  temp.results;
by comcd;
retain ;
if first. comcd then  do;
CsrcIcCd1_1= CsrcIcCd1;
csrcicnm1_1= csrcicnm1;
end;

else if CsrcIcCd1^=''  then  do;
CsrcIcCd1_1= CsrcIcCd1;
csrcicnm1_1= csrcicnm1;
End;

CsrcIcCd1= CsrcIcCd1_1;
csrcicnm1= csrcicnm1_1;
Drop CsrcIcCd1_1 csrcicnm1_1;
If CsrcIcCd1 in ('D' ,'F')  then IndCntrlID=1;
Else IndCntrlID=0;
Label IndCntrlID='IndCntrlID行业管制(虚拟变量)';
Run;

Proc sort data= temp.results;
By comcd year;
Run;

PROC EXPORT DATA= TEMP.Results 
            OUTFILE= "C:\Documents and Settings\Administrator\桌面\Resul
ts.xls" 
            DBMS=EXCEL LABEL REPLACE;
     SHEET="Sheet"; 
     NEWFILE=YES;
RUN;

注：数据结果中，如果在2006至2012年之间该公司股票上市发行或者退市，那么在未上市交易的年份里会出现部分数据为空。
如：2006至2012年之间退市的股票。
上市公司代码 退市日期
C600001	2009-12-29
C600002	2006-04-24
C600003	2010-02-26

2006年后上市的股票：
C601339
C601388
C601515等。

基于上述原因，删除 LiabilitiesTermStruct='债务期限结构LiabilitiesTermStruct'为缺失的数据。
注：财务数据中，分子缺失都记为0。
注：需要无缺失值的标识变量CsrcIcCd1 csrcicnm1  Listedstate最后处理。



